E02DCF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

E02DCF computes a bicubic spline approximation to a set of data values, given on a rectangular grid in the x-y plane. The knots of the spline are located automatically, but a single parameter must be specified to control the trade-off between closeness of fit and smoothness of fit.

2 Specification

```
 SUBROUTINE E02DCF(START, MX, X, MY, Y, F, S, NXEST, NYEST, NX,
1                  LAMDA, NY, MU, C, FP, WRK, LWRK, IWRK, LIWRK,
2                  IFAIL)
 INTEGER           MX, MY, NXEST, NYEST, NX, NY, LWRK, IWRK(LIWRK),
1                  LIWRK, IFAIL
 real              X(MX), Y(MY), F(MX*MY), S, LAMDA(NXEST),
1                  MU(NYEST), C((NXEST-4)*(NYEST-4)), FP, WRK(LWRK)
 CHARACTER*1       START
```

3 Description

This routine determines a smooth bicubic spline approximation $s(x,y)$ to the set of data points $(x_q, y_r, f_{q,r})$, for $q = 1,2,\ldots,m_x$ and $r = 1,2,\ldots,m_y$.

The spline is given in the B-spline representation

$$s(x,y) = \sum_{i=1}^{n_x-4}\sum_{j=1}^{n_y-4} c_{ij} M_i(x) N_j(y), \tag{1}$$

where $M_i(x)$ and $N_j(y)$ denote normalised cubic B-splines, the former defined on the knots λ_i to λ_{i+4} and the latter on the knots μ_j to μ_{j+4}. For further details, see Hayes and Halliday [4] for bicubic splines and de Boor [1] for normalised B-splines.

The total numbers n_x and n_y of these knots and their values $\lambda_1,\ldots,\lambda_{n_x}$ and $\mu_1,\ldots,\mu_{n_y}$ are chosen automatically by the routine. The knots $\lambda_5,\ldots,\lambda_{n_x-4}$ and $\mu_5,\ldots,\mu_{n_y-4}$ are the interior knots; they divide the approximation domain $[x_1, x_{m_x}]\times[y_1, y_{m_y}]$ into $(n_x-7)\times(n_y-7)$ subpanels $[\lambda_i,\lambda_{i+1}]\times[\mu_j,\mu_{j+1}]$, for $i = 4,5,\ldots,n_x-4$, $j = 4,5,\ldots,n_y-4$. Then, much as in the curve case (see E02BEF), the coefficients c_{ij} are determined as the solution of the following constrained minimization problem:

minimize

$$\eta, \tag{2}$$

subject to the constraint

$$\theta = \sum_{q=1}^{m_x}\sum_{r=1}^{m_y} \epsilon_{q,r}^2 \leq S, \tag{3}$$

where η is a measure of the (lack of) smoothness of $s(x,y)$. Its value depends on the discontinuity jumps in $s(x,y)$ across the boundaries of the subpanels. It is zero only when there are no discontinuities and is positive otherwise, increasing with the size of the jumps (see Dierckx [2] for details).

$\epsilon_{q,r}$ denotes the residual $f_{q,r} - s(x_q, y_r)$,

and S is a non-negative number to be specified by the user.

By means of the parameter S, 'the smoothing factor', the user will then control the balance between smoothness and closeness of fit, as measured by the sum of squares of residuals in (3). If S is too large, the spline will be too smooth and signal will be lost (underfit); if S is too small, the spline will pick up too much noise (overfit). In the extreme cases the routine will return an interpolating spline ($\theta = 0$) if S is set to zero, and the least-squares bicubic polynomial ($\eta = 0$) if S is set very large. Experimenting with S-values between these two extremes should result in a good compromise. (See Section 8.3 for advice on choice of S.)

The method employed is outlined in Section 8.5 and fully described in Dierckx [2] and [3]. It involves an adaptive strategy for locating the knots of the bicubic spline (depending on the function underlying the data and on the value of S), and an iterative method for solving the constrained minimization problem once the knots have been determined.

Values of the computed spline can subsequently be computed by calling E02DEF or E02DFF as described in Section 8.6.

4 References

[1] de Boor C (1972) On calculating with B-splines *J. Approx. Theory* **6** 50–62

[2] Dierckx P (1982) A fast algorithm for smoothing data on a rectangular grid while using spline functions *SIAM J. Numer. Anal.* **19** 1286–1304

[3] Dierckx P (1981) An improved algorithm for curve fitting with spline functions *Report TW54* Department of Computer Science, Katholieke Universiteit Leuven

[4] Hayes J G and Halliday J (1974) The least-squares fitting of cubic spline surfaces to general data sets *J. Inst. Math. Appl.* **14** 89–103

[5] Reinsch C H (1967) Smoothing by spline functions *Numer. Math.* **10** 177–183

5 Parameters

1: START — CHARACTER*1 *Input*

On entry: START must be set to 'C' or 'W'.

If START = 'C' (Cold start), the routine will build up the knot set starting with no interior knots. No values need be assigned to the parameters NX, NY, LAMDA, MU, WRK or IWRK.

If START = 'W' (Warm start), the routine will restart the knot-placing strategy using the knots found in a previous call of the routine. In this case, the parameters NX, NY, LAMDA, MU, WRK and IWRK must be unchanged from that previous call. This warm start can save much time in searching for a satisfactory value of S.

Constraint: START = 'C' or 'W'.

2: MX — INTEGER *Input*

On entry: m_x, the number of grid points along the x axis.

Constraint: MX ≥ 4.

3: X(MX) — ***real*** array *Input*

On entry: X(q) must be set to x_q, the x co-ordinate of the qth grid point along the x axis, for $q = 1, 2, \ldots, m_x$.

Constraint: $x_1 < x_2 < \ldots < x_{m_x}$.

4: MY — INTEGER *Input*

On entry: m_y, the number of grid points along the y axis.

Constraint: MY ≥ 4.

5: Y(MY) — ***real*** array *Input*

On entry: Y(r) must be set to y_r, the y co-ordinate of the rth grid point along the y axis, for $r = 1, 2, \ldots, m_y$.

Constraint: $y_1 < y_2 < \ldots < y_{m_y}$.

6: F(MX∗MY) — ***real*** array *Input*

On entry: F($m_y \times (q-1) + r$) must contain the data value $f_{q,r}$, for $q = 1, 2, \ldots, m_x$ and $r = 1, 2, \ldots, m_y$.

7: S — ***real*** *Input*

On entry: the smoothing factor, S.

If S = 0.0, the routine returns an interpolating spline.

If S is smaller than ***machine precision***, it is assumed equal to zero.

For advice on the choice of S, see Section 3 and Section 8.3.

Constraint: S $\geq$ 0.0.

8: NXEST — INTEGER *Input*

9: NYEST — INTEGER *Input*

On entry: an upper bound for the number of knots n_x and n_y required in the x- and y-directions respectively.

In most practical situations, NXEST = $m_x/2$ and NYEST = $m_y/2$ is sufficient. NXEST and NYEST never need to be larger than $m_x + 4$ and $m_y + 4$ respectively, the numbers of knots needed for interpolation (S = 0.0). See also Section 8.4.

Constraint: NXEST $\geq$ 8 and NYEST $\geq$ 8.

10: NX — INTEGER *Input/Output*

On entry: if the warm start option is used, the value of NX must be left unchanged from the previous call.

On exit: the total number of knots, n_x, of the computed spline with respect to the x variable.

11: LAMDA(NXEST) — ***real*** array *Input/Output*

On entry: if the warm start option is used, the values LAMDA(1), LAMDA(2),...,LAMDA(NX) must be left unchanged from the previous call.

On exit: LAMDA contains the complete set of knots λ_i associated with the x variable, i.e., the interior knots LAMDA(5), LAMDA(6),...,LAMDA(NX− 4) as well as the additional knots

$$\text{LAMDA}(1) = \text{LAMDA}(2) = \text{LAMDA}(3) = \text{LAMDA}(4) = \text{X}(1)$$

and

$$\text{LAMDA}(\text{NX}-3) = \text{LAMDA}(\text{NX}-2) = \text{LAMDA}(\text{NX}-1) = \text{LAMDA}(\text{NX}) = \text{X}(\text{MX})$$

needed for the B-spline representation.

12: NY — INTEGER *Input/Output*

On entry: if the warm start option is used, the value of NY must be left unchanged from the previous call.

On exit: the total number of knots, n_y, of the computed spline with respect to the y variable.

13: MU(NYEST) — ***real*** array *Input/Output*

On entry: if the warm start option is used, the values MU(1), MU(2),...,MU(NY) must be left unchanged from the previous call.

On exit: MU contains the complete set of knots μ_i associated with the y variable, i.e., the interior knots MU(5), MU(6),...,MU(NY − 4) as well as the additional knots

$$\mathrm{MU}(1) = \mathrm{MU}(2) = \mathrm{MU}(3) = \mathrm{MU}(4) = \mathrm{Y}(1)$$

and

$$\mathrm{MU}(\mathrm{NY}-3) = \mathrm{MU}(\mathrm{NY}-2) = \mathrm{MU}(\mathrm{NY}-1) = \mathrm{MU}(\mathrm{NY}) = \mathrm{Y}(\mathrm{MY})$$

needed for the B-spline representation.

14: C((NXEST−4)*(NYEST−4)) — ***real*** array *Output*

On exit: the coefficients of the spline approximation. C($(n_y - 4) \times (i - 1) + j$) is the coefficient c_{ij} defined in Section 3.

15: FP — ***real*** *Output*

On exit: the sum of squared residuals, θ, of the computed spline approximation. If FP = 0.0, this is an interpolating spline. FP should equal S within a relative tolerance of 0.001 unless NX = NY = 8, when the spline has no interior knots and so is simply a bicubic polynomial. For knots to be inserted, S must be set to a value below the value of FP produced in this case.

16: WRK(LWRK) — ***real*** array *Workspace*

On entry: if the warm start option is used, the values WRK(1),...,WRK(4) must be left unchanged from the previous call.

This array is used as workspace.

17: LWRK — INTEGER *Input*

On entry: the dimension of the array WRK as declared in the (sub)program from which E02DCF is called.

Constraint: LWRK ≥ 4 × (MX+MY) +11 × (NXEST+NYEST) + NXEST × MY + max(MY,NXEST) + 54.

18: IWRK(LIWRK) — INTEGER array *Workspace*

On entry: if the warm start option is used, the values IWRK(1),...,IWRK(3) must be left unchanged from the previous call.

This array is used as workspace.

19: LIWRK — INTEGER *Input*

On entry: the dimension of the array IWRK as declared in the (sub)program from which E02DCF is called.

Constraint: LIWRK ≥ 3 + MX + MY + NXEST + NYEST.

20: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors detected by the routine:

IFAIL = 1

On entry, START ≠ 'C' or 'W',
or MX < 4,
or MY < 4,
or S < 0.0,
or S = 0.0 and NXEST < MX + 4,
or S = 0.0 and NYEST < MY + 4,
or NXEST < 8,
or NYEST < 8,
or LWRK < 4 × (MX+MY) + 11 × (NXEST+NYEST) + NXEST × MY + max(MY,NXEST) + 54,
or LIWRK < 3 + MX + MY + NXEST + NYEST.

IFAIL = 2

The values of X(q), for $q = 1, 2, \ldots$,MX, are not in strictly increasing order.

IFAIL = 3

The values of Y(r), for $r = 1, 2, \ldots$,MY, are not in strictly increasing order.

IFAIL = 4

The number of knots required is greater than allowed by NXEST and NYEST. Try increasing NXEST and/or NYEST and, if necessary, supplying larger arrays for the parameters LAMDA, MU, C, WRK and IWRK. However, if NXEST and NYEST are already large, say NXEST > MX/2 and NYEST > MY/2, then this error exit may indicate that S is too small.

IFAIL = 5

The iterative process used to compute the coefficients of the approximating spline has failed to converge. This error exit may occur if S has been set very small. If the error persists with increased S, consult NAG.

If IFAIL = 4 or 5, a spline approximation is returned, but it fails to satisfy the fitting criterion (see (2) and (3) in Section 3) – perhaps by only a small amount, however.

7 Accuracy

On successful exit, the approximation returned is such that its sum of squared residuals FP is equal to the smoothing factor S, up to a specified relative tolerance of 0.001 – except that if $n_x = 8$ and $n_y = 8$, FP may be significantly less than S: in this case the computed spline is simply the least-squares bicubic polynomial approximation of degree 3, i.e., a spline with no interior knots.

8 Further Comments

8.1 Timing

The time taken for a call of E02DCF depends on the complexity of the shape of the data, the value of the smoothing factor S, and the number of data points. If E02DCF is to be called for different values of S, much time can be saved by setting START = 'W' after the first call.

8.2 Weighting of Data Points

E02DCF does not allow individual weighting of the data values. If these were determined to widely differing accuracies, it may be better to use E02DDF. The computation time would be very much longer, however.

8.3 Choice of S

If the standard deviation of $f_{q,r}$ is the same for all q and r (the case for which this routine is designed – see Section 8.2.) and known to be equal, at least approximately, to σ, say, then following Reinsch [5] and choosing the smoothing factor S in the range $\sigma^2(m \pm \sqrt{2m})$, where $m = m_x m_y$, is likely to give a good start in the search for a satisfactory value. If the standard deviations vary, the sum of their squares over all the data points could be used. Otherwise experimenting with different values of S will be required from the start, taking account of the remarks in Section 3.

In that case, in view of computation time and memory requirements, it is recommended to start with a very large value for S and so determine the least-squares bicubic polynomial; the value returned for FP, call it FP_0, gives an upper bound for S. Then progressively decrease the value of S to obtain closer fits – say by a factor of 10 in the beginning, i.e., $S = \mathrm{FP}_0/10$, $S = \mathrm{FP}_0/100$, and so on, and more carefully as the approximation shows more details.

The number of knots of the spline returned, and their location, generally depend on the value of S and on the behaviour of the function underlying the data. However, if E02DCF is called with START = 'W', the knots returned may also depend on the smoothing factors of the previous calls. Therefore if, after a number of trials with different values of S and START = 'W', a fit can finally be accepted as satisfactory, it may be worthwhile to call E02DCF once more with the selected value for S but now using START = 'C'. Often, E02DCF then returns an approximation with the same quality of fit but with fewer knots, which is therefore better if data reduction is also important.

8.4 Choice of NXEST and NYEST

The number of knots may also depend on the upper bounds NXEST and NYEST. Indeed, if at a certain stage in E02DCF the number of knots in one direction (say n_x) has reached the value of its upper bound (NXEST), then from that moment on all subsequent knots are added in the other (y) direction. Therefore the user has the option of limiting the number of knots the routine locates in any direction. For example, by setting NXEST = 8 (the lowest allowable value for NXEST), the user can indicate that he wants an approximation which is a simple cubic polynomial in the variable x.

8.5 Outline of Method Used

If $S = 0$, the requisite number of knots is known in advance, i.e., $n_x = m_x + 4$ and $n_y = m_y + 4$; the interior knots are located immediately as $\lambda_i = x_{i-2}$ and $\mu_j = y_{j-2}$, for $i = 5, 6, \ldots, n_x - 4$ and $j = 5, 6, \ldots, n_y - 4$. The corresponding least-squares spline is then an interpolating spline and therefore a solution of the problem.

If $S > 0$, suitable knot sets are built up in stages (starting with no interior knots in the case of a cold start but with the knot set found in a previous call if a warm start is chosen). At each stage, a bicubic spline is fitted to the data by least-squares, and θ, the sum of squares of residuals, is computed. If $\theta > S$, new knots are added to one knot set or the other so as to reduce θ at the next stage. The new knots are located in intervals where the fit is particularly poor, their number depending on the value of S and on the progress made so far in reducing θ. Sooner or later, we find that $\theta \leq S$ and at that point the knot sets are accepted. The routine then goes on to compute the (unique) spline which has these knot sets and which satisfies the full fitting criterion specified by (2) and (3). The theoretical solution has $\theta = S$. The routine computes the spline by an iterative scheme which is ended when $\theta = S$ within a relative tolerance of 0.001. The main part of each iteration consists of a linear least-squares computation of special form, done in a similarly stable and efficient manner as in E02BAF for least-squares curve fitting.

An exception occurs when the routine finds at the start that, even with no interior knots ($n_x = n_y = 8$), the least-squares spline already has its sum of residuals $\leq S$. In this case, since this spline (which is simply a bicubic polynomial) also has an optimal value for the smoothness measure η, namely zero, it is returned at once as the (trivial) solution. It will usually mean that S has been chosen too large.

For further details of the algorithm and its use see Dierckx [2].

8.6 Evaluation of Computed Spline

The values of the computed spline at the points (TX(r),TY(r)), for $r = 1, 2, \ldots$,N, may be obtained in the ***real*** array FF, of length at least N, by the following code:

```
IFAIL = 0
CALL E02DEF(N,NX,NY,TX,TY,LAMDA,MU,C,FF,WRK,IWRK,IFAIL)
```

where NX, NY, LAMDA, MU and C are the output parameters of E02DCF, WRK is a ***real*** workspace array of length at least NY -4, and IWRK is an integer workspace array of length at least NY -4.

To evaluate the computed spline on a KX by KY rectangular grid of points in the x-y plane, which is defined by the x co-ordinates stored in TX(q), for $q = 1, 2, \ldots$,KX, and the y co-ordinates stored in TY(r), for $r = 1, 2, \ldots$,KY, returning the results in the ***real*** array FG which is of length at least KX $\times$ KY, the following call may be used:

```
IFAIL = 0
CALL E02DFF(KX,KY,NX,NY,TX,TY,LAMDA,MU,C,FG,WRK,LWRK,
*            IWRK,LIWRK,IFAIL)
```

where NX, NY, LAMDA, MU and C are the output parameters of E02DCF, WRK is a ***real*** workspace array of length at least LWRK = min(NWRK1,NWRK2), NWRK1 = KX $\times$ 4 + NX, NWRK2 = KY $\times$ 4 + NY, and IWRK is an integer workspace array of length at least LIWRK = KY + NY $-$ 4 if NWRK1 $\geq$ NWRK2, or KX + NX $-$ 4 otherwise. The result of the spline evaluated at grid point (q, r) is returned in element (KY$\times(q-1)+r$) of the array FG.

9 Example

This example program reads in values of MX, MY, x_q, for $q = 1, 2, \ldots$,MX, and y_r, for $r = 1, 2, \ldots$,MY, followed by values of the ordinates $f_{q,r}$ defined at the grid points (x_q,y_r). It then calls E02DCF to compute a bicubic spline approximation for one specified value of S, and prints the values of the computed knots and B-spline coefficients. Finally it evaluates the spline at a small sample of points on a rectangular grid.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     E02DCF Example Program Text
*     Mark 18 Revised.  NAG Copyright 1997.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MXMAX, MYMAX, NXEST, NYEST, NMAX, LWRK, LIWRK
      PARAMETER        (MXMAX=11,MYMAX=9,NXEST=MXMAX+4,NYEST=MYMAX+4,
     +                 NMAX=7,LWRK=4*(MXMAX+MYMAX)+11*(NXEST+NYEST)
     +                 +NXEST*MYMAX+NXEST+54,LIWRK=3+MXMAX+MYMAX+NXEST+
     +                 NYEST)
*     .. Local Scalars ..
      real             DELTA, FP, S, XHI, XLO, YHI, YLO
      INTEGER          I, IFAIL, J, MX, MY, NPX, NPY, NX, NY
      CHARACTER        START
*     .. Local Arrays ..
      real             C((NXEST-4)*(NYEST-4)), F(MXMAX*MYMAX),
     +                 FG(NMAX*NMAX), LAMDA(NXEST), MU(NYEST), PX(NMAX),
     +                 PY(NMAX), WRK(LWRK), X(MXMAX), Y(MYMAX)
      INTEGER          IWRK(LIWRK)
```

```
*     .. External Subroutines ..
      EXTERNAL         CPRINT, E02DCF, E02DFF
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'E02DCF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
*     Input the number of X, Y co-ordinates MX, MY.
      READ (NIN,*) MX, MY
      IF (MX.GT.0 .AND. MX.LE.MXMAX .AND. MY.GT.0 .AND. MY.LE.MYMAX)
     +    THEN
*        Input the X co-ordinates followed by the Y co-ordinates.
         READ (NIN,*) (X(I),I=1,MX)
         READ (NIN,*) (Y(I),I=1,MY)
*        Input the MX*MY function values F at the grid points.
         READ (NIN,*) (F(I),I=1,MX*MY)
         START = 'Cold Start'
         READ (NIN,*,END=100) S
*        Determine the spline approximation.
         IFAIL = 0
*
         CALL E02DCF(START,MX,X,MY,Y,F,S,NXEST,NYEST,NX,LAMDA,NY,MU,C,
     +              FP,WRK,LWRK,IWRK,LIWRK,IFAIL)
*
         WRITE (NOUT,*)
         WRITE (NOUT,99999) 'Calling with smoothing factor S =', S,
     +     ': NX =', NX, ', NY =', NY, '.'
         WRITE (NOUT,*)
         WRITE (NOUT,*)
     +     '               I   Knot LAMDA(I)      J     Knot MU(J)'
         WRITE (NOUT,*)
         DO 20 J = 4, MAX(NX,NY) - 3
            IF (J.LE.NX-3 .AND. J.LE.NY-3) THEN
               WRITE (NOUT,99997) J, LAMDA(J), J, MU(J)
            ELSE IF (J.LE.NX-3) THEN
               WRITE (NOUT,99997) J, LAMDA(J)
            ELSE IF (J.LE.NY-3) THEN
               WRITE (NOUT,99996) J, MU(J)
            END IF
   20    CONTINUE
         CALL CPRINT(C,NY,NX,NOUT)
         WRITE (NOUT,*)
         WRITE (NOUT,99998) 'Sum of squared residuals FP =', FP
         IF (FP.EQ.0.0e+0) THEN
            WRITE (NOUT,*) '(The spline is an interpolating spline)'
         ELSE IF (NX.EQ.8 .AND. NY.EQ.8) THEN
            WRITE (NOUT,*)
     +        '(The spline is the least-squares bi-cubic polynomial)'
         END IF
*        Evaluate the spline on a rectangular grid at NPX*NPY points
*        over the domain (XLO to XHI) x (YLO to YHI).
         READ (NIN,*) NPX, XLO, XHI
         READ (NIN,*) NPY, YLO, YHI
         IF (NPX.LE.NMAX .AND. NPY.LE.NMAX) THEN
            DELTA = (XHI-XLO)/(NPX-1)
            DO 40 I = 1, NPX
```

```
               PX(I) = MIN(XLO+(I-1)*DELTA,XHI)
   40       CONTINUE
            DO 60 I = 1, NPY
               PY(I) = MIN(YLO+(I-1)*DELTA,YHI)
   60       CONTINUE
*
            CALL E02DFF(NPX,NPY,NX,NY,PX,PY,LAMDA,MU,C,FG,WRK,LWRK,IWRK,
     +                  LIWRK,IFAIL)
*
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Values of computed spline:'
            WRITE (NOUT,*)
            WRITE (NOUT,99995) '          X', (PX(I),I=1,NPX)
            WRITE (NOUT,*) '     Y'
            DO 80 I = NPY, 1, -1
               WRITE (NOUT,99994) PY(I), (FG(NPY*(J-1)+I),J=1,NPX)
   80       CONTINUE
         END IF
      END IF
  100 CONTINUE
      STOP
*
99999 FORMAT (1X,A,1P,e13.4,A,I5,A,I5,A)
99998 FORMAT (1X,A,1P,e13.4)
99997 FORMAT (1X,I16,F12.4,I11,F12.4)
99996 FORMAT (1X,I39,F12.4)
99995 FORMAT (1X,A,7F8.2)
99994 FORMAT (1X,F8.2,3X,7F8.2)
      END
*
      SUBROUTINE CPRINT(C,NY,NX,NOUT)
*     .. Scalar Arguments ..
      INTEGER           NOUT, NX, NY
*     .. Array Arguments ..
      real              C(NY-4,NX-4)
*     .. Local Scalars ..
      INTEGER           I, J
*     .. Executable Statements ..
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'The B-spline coefficients:'
      WRITE (NOUT,*)
      DO 20 I = 1, NY - 4
         WRITE (NOUT,99999) (C(I,J),J=1,NX-4)
   20 CONTINUE
      RETURN
*
99999 FORMAT (1X,8F9.4)
      END
```

9.2 Program Data

```
E02DCF Example Program Data
 11    9      MX, MY, number of grid points on the X and Y axes
 0.0000E+00  5.0000E-01  1.0000E+00  1.5000E+00  2.0000E+00
 2.5000E+00  3.0000E+00  3.5000E+00  4.0000E+00  4.5000E+00
 5.0000E+00   End of MX grid points
 0.0000E+00  5.0000E-01  1.0000E+00  1.5000E+00  2.0000E+00
 2.5000E+00  3.0000E+00  3.5000E+00  4.0000E+00   End of MY grid points
```

```
 1.0000E+00  8.8758E-01  5.4030E-01  7.0737E-02 -4.1515E-01
-8.0114E-01 -9.7999E-01 -9.3446E-01 -6.5664E-01  1.5000E+00
 1.3564E+00  8.2045E-01  1.0611E-01 -6.2422E-01 -1.2317E+00
-1.4850E+00 -1.3047E+00 -9.8547E-01  2.0600E+00  1.7552E+00
 1.0806E+00  1.5147E-01 -8.3229E-01 -1.6023E+00 -1.9700E+00
-1.8729E+00 -1.4073E+00  2.5700E+00  2.1240E+00  1.3508E+00
 1.7684E-01 -1.0404E+00 -2.0029E+00 -2.4750E+00 -2.3511E+00
-1.6741E+00  3.0000E+00  2.6427E+00  1.6309E+00  2.1221E-01
-1.2484E+00 -2.2034E+00 -2.9700E+00 -2.8094E+00 -1.9809E+00
 3.5000E+00  3.1715E+00  1.8611E+00  2.4458E-01 -1.4565E+00
-2.8640E+00 -3.2650E+00 -3.2776E+00 -2.2878E+00  4.0400E+00
 3.5103E+00  2.0612E+00  2.8595E-01 -1.6946E+00 -3.2046E+00
-3.9600E+00 -3.7958E+00 -2.6146E+00  4.5000E+00  3.9391E+00
 2.4314E+00  3.1632E-01 -1.8627E+00 -3.6351E+00 -4.4550E+00
-4.2141E+00 -2.9314E+00  5.0400E+00  4.3879E+00  2.7515E+00
 3.5369E-01 -2.0707E+00 -4.0057E+00 -4.9700E+00 -4.6823E+00
-3.2382E+00  5.5050E+00  4.8367E+00  2.9717E+00  3.8505E-01
-2.2888E+00 -4.4033E+00 -5.4450E+00 -5.1405E+00 -3.5950E+00
 6.0000E+00  5.2755E+00  3.2418E+00  4.2442E-01 -2.4769E+00
-4.8169E+00 -5.9300E+00 -5.6387E+00 -3.9319E+00   End of data values
 0.1         S, smoothing factor
 6    0.0   5.0
 5    0.0   4.0
```

9.3 Program Results

```
E02DCF Example Program Results

Calling with smoothing factor S =   1.0000E-01: NX =   10, NY =   13.

               I   Knot LAMDA(I)      J     Knot MU(J)

               4      0.0000          4      0.0000
               5      1.5000          5      1.0000
               6      2.5000          6      2.0000
               7      5.0000          7      2.5000
                                      8      3.0000
                                      9      3.5000
                                     10      4.0000

The B-spline coefficients:

   0.9918   1.5381   2.3913   3.9845   5.2138   5.9965
   1.0546   1.5270   2.2441   4.2217   5.0860   6.0821
   0.6098   0.9557   1.5587   2.3458   3.3860   3.7716
  -0.2915  -0.4199  -0.7399  -1.1763  -1.5527  -1.7775
  -0.8476  -1.3296  -1.8521  -3.3468  -4.3628  -5.0085
  -1.0168  -1.5952  -2.4022  -3.9390  -5.4680  -6.1656
  -0.9529  -1.3381  -2.2844  -3.9559  -5.0032  -5.8709
  -0.7711  -1.0914  -1.8488  -3.2549  -3.9444  -4.7297
  -0.6476  -1.0373  -1.5936  -2.5887  -3.3485  -3.9330

Sum of squared residuals FP =   1.0004E-01
```

```
Values of computed spline:

          X    0.00    1.00    2.00    3.00    4.00    5.00
 Y
4.00       -0.65   -1.36   -1.99   -2.61   -3.25   -3.93
3.00       -0.98   -1.97   -2.91   -3.91   -4.97   -5.92
2.00       -0.42   -0.83   -1.24   -1.66   -2.08   -2.48
1.00        0.54    1.09    1.61    2.14    2.71    3.24
0.00        0.99    2.04    3.03    4.01    5.02    6.00
```